MATHEMATICS 201-105-RE
Linear Algebra
Martin Huard
Winter 2006

Vectors, Lines and Planes with Maple

When working with vectors on Maple, the first thing to do is to load the *linalg* package, which contains a lot of the commands that we need.

> **with(linalg):**

Vectors

Definition of Vectors with Maple

Vectors in Maple can be defined as points using [ ] brackets. For example, the vector $\vec{u} = (1,-2,4)$ is defined by

> u:=[1,-2,4];

Note that you can also define a vector with the command **vector([1,-2,4])**, or using the vector palette on you left entering the entries one at a time and pressing the TAB key to move to the next entry.

Plotting Vectors

We can represent vectors graphically using the **arrow** command from the *plottools* package and the **display** command from the *plots* package.

> with(plottools):
with(plots):

The **arrow** command has the following format:

arrow(base, dir , wb, wh, hh)

base - base of the arrow, that is the point where the arrow starts.
dir - either the tip of the arrow, or the direction (written as vector(v), where v is a point or vector)
wb - width of the body of the arrow
wh - width of the head of the arrow
hh - height of the head of the arrow as a ratio of the length of the body

For example, let us plot the vector $\vec{u} = (1,-2,4)$.

```
> zero:=[0,0,0]:
  gu:=arrow(zero,u,0.1,0.3,0.1):
  display(gu, axes=normal, labels=[x,y,z]);
```

The angle of vision can be moved by clicking on the graph and moving your mouse (with the button still pressed down). Try it!

Operations with vectors

Maple can handle most of the common operations with vectors. For example, suppose we have the points $A(2,-3,1)$, $B(2,-1,5)$ and $C(-1,-3,4)$.

> A:=[2,-3,1];
B:=[2,-1,5];
C:=[-1,-3,4];

Then the vectors $\overrightarrow{AB}$ and $\overrightarrow{AC}$ are given by

> AB:=B-A;
AC:=C-A;

Vector Addition

The addition of vectors is done as expected. For example, $\overrightarrow{AB}+\overrightarrow{AC}$ is given by

> AB+AC;

Multiplication by a scalar

The vector $4\overrightarrow{AB}$ is given by

> 4*AB;

Magnitude of a vector

Maple has the command **norm** in the *linalg* package to find the magnitude of a vector. For example, let us find $\left\|\overrightarrow{AB}\right\|$

> norm(A,2);

Thus, $\left\|\overrightarrow{AB} =\right\| = 2\sqrt{5}$.

Note: the 2 in the command means that we take a square root. Were we to put a four, that is **norm(AB,4)**, then we would be evaluating $\sqrt[4]{0^4+2^4+4^4}$, which does not correspond to the magnitude as usually defined. Ergo, just always use a 2.

Scalar or dot product

The command for the scalar product of two vectors is **dotprod** taken from the *linalg* package.

> dotprod(AB,AC);

Hence, $\overrightarrow{AB}\bullet\overrightarrow{AC} = (0,2,4)\bullet(-3,0,3) = 12$.

Vector or cross product

For the vector product, we use the **crossprod** command from the *linalg* package.

> crossprod(AB,AC);

Hence $\overrightarrow{AB}\times\overrightarrow{AC} = (0,2,4)\times(-3,0,3) = (6,-12,6)$

Angle between two vectors

We can find the angle between two vectors using the angle command. For example, let us find the angle between the vectors $\overrightarrow{AB}$ and $\overrightarrow{AC}$.

> theta:=angle(AB,AC);

To obtain a decimal answer, we use the **evalf** command.

> evalf(theta);

Note that Maple gives the answer in radians. To have the angle in degrees, we use the **convert** command.

> convert(theta,degrees); evalf(%);

Hence $\theta \approx 50.8°$.

Lines

We can plot lines in $\mathbb{R}^3$ with Maple when the line is given in vector form. For example, if we have the line $l:(x,y,z)=(1,3,-5)+t(2,-3,4)$, then we can define it

> l:=[1,3,-5]+t*[2,-3,4];

and use the command **spacecurve** from the *plots* package to graph it.

```
> with(plots):
  gline:=spacecurve(evalm(l),t=-3..3):
  display(gline, axes=normal, labels=[x,y,z],scaling=constrained);
```

Planes

Vector Form

To plot a plane in vector form, such as $\pi:(x,y,z)=(2,-3,1)+s(0,2,4)+t(-3,0,3)$ we start by defining it,

> pi:= [2, -3, 1]+s*[0, 2, 4]+t*[-3, 0, 3];

and then use the **plot3d** command.

```
> gplane:=plot3d(evalm(pi),t=-2..2,s=-2..2):
  display(gplane, axes=normal, labels=[x,y,z]);
```

General Form

To plot a plane in general form , such as $\pi_2:5x-2y-z=5$, we have

> pi2:=5*x-2*y-z=5;

and use the **implicitplot3d** command.

```
> gplane2:=implicitplot3d(pi2,x=-8..8,y=-8..8,z=-8..8):
> display(gplane2, axes=normal, labels=[x,y,z]);
```

To plot more than one plane (or line or vector), simply add to the "display" command.

> display(gline,gplane, axes=normal, labels=[x,y,z]);